

# Maple Code

October 23, 2023 1:42 PM

```
> `type/unipoly` := proc(f) type(f, list([integer, nonnegint])) end:
> AddPol := proc( f::unipoly, g::unipoly ) local m,n,h,i,j,k,c;
  m,n := nops(f), nops(g);
  h := Array(1..m+n);
  i,j,k := 1,1,0;
  while i<=m and j<=n do
    if f[i][2]>g[j][2] then k++; h[k] := f[i]; i++;
    elif f[i][2]<g[j][2] then k++; h[k] := g[j]; j++;
    else c := f[i][1]+g[j][1];
      if c>0 then k++; h[k] := [c, f[i][2]]; fi;
      i++; j++;
    fi;
  od;
  while i<=m do k++; h[k] := f[i]; i++; od;
  while j<=n do k++; h[k] := g[j]; j++; od;
  h := [seq( h[i], i=1..k )];
end:
```

← Copy fi into h-

```
> MAPLE2SPARSE := proc(f,x) local cofs,mons,i;
  cofs := coeffs(f,x,'mons'); cofs,mons := [cofs],[mons];
  [seq( [cofs[i],degree(mons[i],x)], i=1..nops(cofs) )];
end:
SPARSE2MAPLE := proc(f,x) local t;
  add( t[1]*x^t[2], t in f );
end:
```

Forget to sort.

```
> f := 2*x^3+4*x+3*x^2;
F := MAPLE2SPARSE(f,x);
```

$$f := 2x^3 + 3x^2 + 4x$$

$$F := [[2, 3], [3, 2], [4, 1]] \quad (1)$$

```
> g := 3*x^4-3*x^2+x^3-2*x;
G := MAPLE2SPARSE(g,x);
```

$$g := 3x^4 + x^3 - 3x^2 - 2x$$

$$G := [[3, 4], [1, 3], [-3, 2], [-2, 1]] \quad (2)$$

```
> H := AddPol(F,G);
SPARSE2MAPLE(H,x);
f+g;
```

$$H := [[3, 4], [3, 3], [2, 1]]$$

$$3x^4 + 3x^3 + 2x$$

$$3x^4 + 3x^3 + 2x \quad (3)$$

```
> MulPol := proc(f::unipoly,g::unipoly) local h,s,t,T;
  h := []; # h = 0
  for s in f do
    T := [seq( [s[1]*t[1], s[2]+t[2]], t in g )];
    h := AddPol(h,T);
  od;
  h;
end:
```

s = fi = [c, e]

$[c, e] \times [[c_1, e_1], [c_2, e_2], \dots, [c_n, e_n]]$

```
> H := MulPol(F,G);
SPARSE2MAPLE(H,x);
expand(f*g);
```

$$H := [[6, 7], [11, 6], [9, 5], [-9, 4], [-18, 3], [-8, 2]]$$

$$6x^7 + 11x^6 + 9x^5 - 9x^4 - 18x^3 - 8x^2$$

$$6x^7 + 11x^6 + 9x^5 - 9x^4 - 18x^3 - 8x^2 \quad (4)$$